Math 2280-1

Wednesday March 3, 2010

4.1, 4.3: First order systems of Differential Equations

Why you should expect existence and uniqueness for the IVP

Example: Consider the initial value problem related to page 4 of this yesterday's class notes.

$$\begin{bmatrix} \frac{dx}{dt} \\ \frac{dy}{dt} \end{bmatrix} = \begin{bmatrix} y \\ -1.01 \ x - 0.2 \ y \end{bmatrix}$$
$$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

Remember, this was a first order system we got by converting a second order DE for a certain damped spring oscillation problem,


$$x'(t) + .2 x'(t) + 1.01 x(t) = 0$$

 $x(0)=1$
 $x'(0)=0$

so we already know the solution. It is (fill in here!)

But let's pretend we don't know the solution, and try to understand why we would expect one, and why we would expect it to be unique, just based on the geometric meaning of this IVP: If we plot the solution curve [(x(t),y(t)]] in the plane, it passes through [1,0] at t=0, and then moves so that its tangent vector [x'(t),y'(t)] agrees with the DE vector field at the point [x(t),y(t)], namely with [y,-1.01x-.2y]. So if we create a picture of the vector field we can visualize where the solution trajectory must start and then go:

- > restart:with(DEtools):with(plots):with(LinearAlgebra):
- > fieldplot([y,-1.01*x-.2*y],x=-1..1,y=-1..1.);



We can use Euler's method to get an approximate numerical solution! In other words, approximate using the tangent line approximation, and for the tangent vector use what the differential equation says it is (those equals signs are really approximate equals signs):

$$\begin{bmatrix} x(t+\Delta t) \\ y(t+\Delta t) \end{bmatrix} = \begin{bmatrix} x(t) \\ y(t) \end{bmatrix} + \Delta t \begin{bmatrix} \frac{d}{dt} x(t) \\ \frac{d}{dt} y(t) \end{bmatrix}$$
$$\begin{bmatrix} x(t+\Delta t) \\ y(t+\Delta t) \end{bmatrix} = \begin{bmatrix} x(t) \\ y(t) \end{bmatrix} + \Delta t \begin{bmatrix} y(t) \\ -1.01 x(t) - 0.2 y(t) \end{bmatrix}.$$

With the specified initial conditions we start at the point [1,0], and then follow the tangent field around a curve which looks a lot like a spiral. If we want to get a numerical approximation, we can just transpose Euler (or improved Euler or Runge–Kutta) for first order equations to first order systems . . . all the original justifications make just as much sense in vector form!!! (Or we can use one of Maple's numerical packages.)

```
> f1:=(x,y,t)->y:
  f2:=(x,y,t)->-1.01*x-.2*y:
    #our right hand side F(X,t)=[f1(X,t),f2(X,t)]
    #in our first order system of DE's dX/dt = F(t,X).
> n:=100:
    #number of time steps
  t0:=0: #initial time
  tn:=evalf(2*Pi):
    #final time choice
  h:=(tn-t0)/n:
    #time step
  xs:=Vector(n+1):
  ys:=Vector(n+1): #vectors to hold Euler approximates
  xs[1]:=1:
  ys[1]:=0:
             #initial conditions
```

Euler loop

```
> #Euler loop:
    for i from 1 to n do
        X:=xs[i]:
        Y:=ys[i]:
        t:=t0+(i-1)*h:
        k1:=f1(X,Y,t):
        k2:=f2(X,Y,t):
        #k1 and k2 are velocity components
        xs[i+1]:=X+h*k1:
        ys[i+1]:=Y+h*k2:
        #update xs and ys with Euler!
    od:
```


Any first order system is analogous to the above example, and that's why we expect existence and uniqueness in general. As with first order DE's, we'll get better approximate solutions using more sophisticated algorithms, like improved Euler and Runge Kutta. For example, here's some improved Euler code, first for the first order scalar ODE, then converted to the system above.


Original improved Euler loop, stolen from the file *numerical1.mw*, in our class lectures directory.

We now vectorize everything, and store the [x,y] values in a vector so we can make a picture like we did for unimproved Euler. (Before executing the code one the next page I re–entered all the initialization definitions from the previous page.)

Improved Euler loop for first order system of two differential equations:

```
> #Improved Euler loop:
  for i from 1 to n do
    X:=xs[i]:
    Y:=ys[i]:
    t:=t0+(i-1)*h:
    k11:=f1(X,Y,t):
    k12:=f2(X,Y,t):
      #k11 and k12 are left-hand tangent estimates
    k21:=f1(X+h*k11,Y+h*k12,t):
    k22:=f2(X+h*k11,Y+h*k12,t):
      #k21 and k22 are right-hand tangent estimates
    xs[i+1]:=X+h*(k11+k21)/2:
    ys[i+1]:=Y+h*(k12+k22)/2:
      #update xs and ys with improved Euler!
  od:
> tangefield:=fieldplot([y,-1.01*x-.2*y],x=-1..1.01,y=-1..1.):
  alpha:=arctan(.1):
  C:=1/cos(alpha):
  Eulerapprox:=pointplot({seq([xs[i],ys[i]],i=1..n+1)}):
  exactsol:=plot([C*exp(-.1*s)*cos(s-alpha),
  C*exp(-.1*s)*(-.1*cos(s-alpha)-sin(s-alpha)),
    s=0..2*Pi],color=black): #This is the actual
    #solution to the IVP, more complicated than
    #the formula in yesterday's notes, but correct
    #for this IVP, I hope.
  display({tangefield, Eulerapprox, exactsol}, title="go with the flow better with
  Improved Euler");
```





Remarks:

- (1) Any single nth order differential equation is equivalent to a related first order system of n first order differential equations, as we've been discussing. You get this system in the same way as we did in the worked out example. Then the existence and uniqueness theorem for first order systems, which makes clear geometric sense, implies the existence and uniqueness theorem that we quoted a month ago for nth order (linear) differential equations (which wasn't as intuitively believable).
- (2) If you want to solve an nth order differential equation numerically you first convert it to a first order system, and then use Euler or Runge–Kutta, to solve it! (Or you use a package that does something like that while you aren't looking.)

Other neat examples:

The pictures we drew above are called "phase portraits", like the 1-d phase diagrams we drew for autonomous single first order equations in chapter 2. (To get pictures analgous to the 2-d (t,x) slope field pictures you would need to introduce a "t"-axis, so you'd be drawing a 3-d (t,x,y) sketch.) If you remember that for the mechanical examples the vertical y-direction corresponds to velocity and the horizontal x-direction corresponds to position, then the phase portrait is a beautiful way to exhibit position and velocity simultaneously (although time is not explicitly shown).

Here are some other mechanical systems converted into first order systems:

Undamped harmonic oscillator:

Let m=k=1 and c=0 in the mass-spring differential equation. Thus, for displacement x(t),

$$\frac{\mathrm{d}^2}{\mathrm{d}t^2} x(t) + x(t) = 0.$$


Converting this to a first order system as before, i.e. introducing the function y(t) as a proxy for x'(t), yields

$$\begin{bmatrix} \frac{dx}{dt} \\ \frac{dy}{dt} \end{bmatrix} = \begin{bmatrix} y \\ -x \end{bmatrix}$$

Can you explain the picture on the next page? Can you explain why the solutions curves [x(t),y(t)] traverse circles, directly from the DE? How is this related to the total energy (KE+PE) of the oscillator being constant?

> restart: with(plots): plot1:=fieldplot([y,-x],x=-1.5..1.5,y=-1.5..1.5): plot2:=plot({[cos(t),-sin(t),t=0..2*Pi], [.5*cos(t),-.5*sin(t),t=0..2*Pi]},color=black): display({plot1,plot2},title='undamped harmonic oscillator');

undamped harmonic oscillator



Nonlinear pendulum: Recall that (for g=L) the non–linear pendulum equation was

$$L\left(\frac{d^2}{dt^2}\,\theta(t)\right) + g\,\sin(\theta(t)) = 0$$

and that we derived this equation by setting the derivative of the total energy equal to zero, where this sum of kinetic and potential energy was given by

$$E = \frac{1 m L^2 \left(\frac{\mathrm{d}}{\mathrm{d}t} \theta(t)\right)^2}{2} + m g L \left(1 - \cos(\theta(t))\right).$$

If we take L=g, and write

$$x = \theta(t)$$
$$y = \frac{d}{dt} \theta(t)$$

we get the equivalent first order system


$$\begin{bmatrix} \frac{dx}{dt} \\ \frac{dy}{dt} \end{bmatrix} = \begin{bmatrix} y \\ -\sin(x) \end{bmatrix}.$$

What are the equilibrium solutions of this autonomous system?

Which ones do you think are "stable"?

Interpret this picture:

nonlinear pendulum motion



>

Notice that if you magnify the picture near the origin, it starts looking like the undamped harmonic oscillator of the previous example. If we added appropriate damping to our pendulum model, then the magnified picture near the origin could conceivably look like our original example of the damped harmonic oscillator. These are geometric illustrations of the idea of linearization! (You could also linearize near the unstable equilibria.)